

Anomalous Wave Dynamics in a Non-Ideal Gas

University of Notre Dame

Department of Aerospace and Mechanical Engineering

Prepared by: Alexander M. Davies

Prepared for: AME 48491 - Undergraduate Research,

Joseph M. Powers

Notre Dame, Indiana 46556

May 11, 2021

Acknowledgements

All of this work would certainly not have been possible without the assistance of Ms. Katherine Pielemeier. Every question and email I ever set her way was responded to with clarity and a true desire to help me learn new and complex material. She is truly one of the brightest individuals I have ever met, and I am forever thankful to have worked with her and that I have been able to learn from one of the best in the field.

I would also like to thank my advisor, Dr. Joseph M. Powers, for his invaluable teaching. My journey to this point in my research career was once a distant dream in which I thought I might never reach, but his teaching and desire to assist me learn pushed me to become a better student, researcher, and individual and got me to where I am today.

Lastly, I would like to thank my mom, my dad, and my sisters for their unending love and support during my research career. Without my family, I would have lost my sanity a long time ago, and I would never have been encouraged to get involved in undergraduate research in the first place, so truly, without them, none of this work would be possible.

Abstract

This report focuses on numerical solutions to shock tube problems in which anomalous behavior is demonstrated in a van der Waals gas. This study aims to explain the physics and provide solutions outlining the phenomenon of discontinuous rarefactions and continuous compressions in regions where isentropes are non-convex in pressure and specific volume space. In this work, the primary solution procedure involves the Richtmyer two-step Lax-Wendroff method to solve the Euler equations governing gas dynamics. For three unique sets of initial conditions, the gas behavior in each regime is identified and analyzed, and the speed of sound is connected to solutions in which anomalous behavior is predicted. It is found that discontinuous rarefactions and isentropic compressions may occur in certain gases near the vicinity of the vapor dome which entails a reversal of the speed of sound's relation to the state variables, and the second law of thermodynamics is ultimately not violated.

Contents

- 1 Introduction 4
- 2 Literature Review 5
- 3 Convexity 8
- 4 Governing Equations 10
- 5 Shock Capturing Methods 13
- 6 The Convex Case 17
- 7 The Non-Convex Case 20
- 8 The Case of Mixed Convexity 22
- 9 Discussions and Conclusions 25
- Appendix A 29

1 Introduction

In this report, anomalous wave behavior exhibited by a van der Waals gas will be explored. The convexity of isentropes in the pressure-specific volume plane will be connected to numerical solutions of standard Sod shock tube problems [1] with van der Waals as the coupled equation of state. A mathematical feature of a function known as convexity will be shown to play a large role in the wave form of solutions, allowing for the possibility of discontinuous rarefactions (often coined “rarefaction shocks” in the literature) and isentropic compressions.

To solve Sod shock tube problems, the Euler equations will be introduced as governing equations. A Sod shock tube problem is one in which a diaphragm separates two given initial conditions (typically one of high pressure and one of low pressure), and the diaphragm is suddenly removed such that waves propagate in both directions. The release of the diaphragm may produce shock waves and rarefaction fans in typical compressible aerodynamics. The Euler equations are a system of non-linear conservation laws for mass, momentum, and energy that explain the flow in the shock tube. A system of conservation laws differs from scalar conservation laws in that dependent variables of each particular equation depend on other dependent variables in the system; thus, solutions for each variable in question (i.e. mass, momentum, and energy in this case) need to be updated simultaneously to provide a solution.

For this work, the second-order Richtmeyer two-step Lax-Wendroff method is used to form solutions, and a comparison to first-order Lax-Friedrichs and an exact solution will explain its efficacy in practice. Numerical effects of the first-order Lax-Friedrichs and second-order Lax-Wendroff methods will be analyzed, and reasoning will be provided for the choice of discretization method before mentioned.

Ultimately, emphasis will be placed on three specific cases of solutions to van der Waals shock tube problems. In the first case, the initial conditions governing the shock-tube problem are located in pressure-specific volume space where isentropes are fully convex. In the

second case, initial conditions lie in the regime where isentropes are fully non-convex, and the third case is a mixed case in which part of the solution space is convex and the other non-convex.

For the remainder of this report, the underlying theory for anomalous wave behavior in the inviscid, compressible, van der Waals gas will be investigated. New concepts such as the fundamental derivative of gas dynamics will be calculated from the adiabatic sound speed in Section 2, the concept of convexity will be explored in Section 3, and implications of the fundamental derivative curve in pressure and specific volume space will be examined on wave dynamics through numerical solutions in Sections 6, 7, and 8 to the governing equations in Section 4.

2 Literature Review

In an investigation of derivatives of isentropes in the pressure and specific volume, $P - v$, plane as they apply to single-phase gases and phase transitions, Bethe [2] theorized the possibility that $\partial^2 P / \partial v^2|_s < 0$, where s is the specific entropy, could be found for single-phase gases with values of $\delta = R/c_v$ less than 0.06, or c_v/R values of 17.5 or greater where R is the specific gas constant $R = \hat{R}/M$ ($\hat{R} = 8.314$ J/mol/K), M is the molecular mass, and c_v is the specific heat at constant volume. However, he noted that such large values of c_v/R in a gas are nearly impossible in nature. Thus, for all gases, the claim was made that $\partial^2 P / \partial v^2|_s > 0$. Zel'dovich [3] conveyed this phenomenon as well; however, he believed that certain gases could demonstrate this behavior. Lambrakis and Thompson [4] were among the first to investigate these particular fluids, and found that a number of fluids in fact possess δ values of less than 0.06. Fluids that exhibit this particular behavior are recognized in the literature as Bethe-Zel'dovich-Thompson fluids, or BZT fluids, coined after the pioneers of the primary discovery.

The adiabatic sound speed, c , is well known to be given by the following formula [5],

$$c^2 = -v^2 \left(\frac{\partial P}{\partial v} \right)_s. \quad (1)$$

Thompson [6], introduced a relation to $\partial^2 P / \partial v^2|_s$ and the sound speed that provides context to the anomalous phenomenon. His relation known as the fundamental derivative, \mathcal{G} , of gas dynamics is given in Eq. (2) by,

$$\mathcal{G} = \frac{v^3}{2c^2} \left(\frac{\partial^2 P}{\partial v^2} \right)_s. \quad (2)$$

Note, the value of the scaling factor $v^3/2c^2$ is always positive. This is of particular importance as the sign of \mathcal{G} directly depends on the second derivative of pressure along the isentrope. Later, Colonna and Guardone [5] in a molecular investigation of non-classical gas dynamics provided an exactly equivalent definition of the fundamental derivative given by:

$$\mathcal{G} = 1 - \frac{v}{c} \left(\frac{\partial c}{\partial v} \right)_s. \quad (3)$$

The exact analysis in this paper could have been computed using Eq. (3); however, analysis here is focused on Eq. (2). As summarized by Colonna and Guardone [5], the reason for which large c_v gases demonstrate negative second derivatives of pressure with respect to specific volume is seen in Eq. (4). From thermodynamics, one has

$$\left(\frac{\partial T}{\partial v} \right)_s = -\frac{T}{c_v} \left(\frac{\partial P}{\partial T} \right)_v. \quad (4)$$

As c_v increases, the closer $\partial T / \partial v|_s$ approaches 0. This entails that temperature, T , variance with specific volume is minimized and tends towards an isentrope. Thus, for equations of state such as van der Waals where isotherms are non-convex in the vicinity of the vapor dome, one could expect non-convex isentropes for high specific heat at constant volume gases. Certain gases with $\delta < 0.06$ are provided in Tab. 1 [5], along with important information

regarding critical constants. Not surprisingly, BZT fluids are typically heavier hydrocarbons and fluorocarbons that possess a large number of degrees of freedom, $N = 2/\delta$, due its large size, mass, and potential kinetic and rotational energy present in the molecule. Another way of demonstrating the necessary condition for a particular fluid to possess a viable $\mathcal{G} < 0$ region is such that its number of degrees of freedom exceeds 35.

Table 1. A summary of BZT gases exhibiting $\delta = R/c_v < 0.06$ and thus a region in which $\partial^2 P/\partial v^2|_s < 0$ for a single-phase vapor.

Fluid	M [g/mol]	P_c [Pa]	T_c [K]	δ	N
Toluene	92.14	3990000	591.8	0.0452	44.3
Octane	114.20	2430000	568.8	0.0273	73.3
Decane	142.29	2060000	617.7	0.0205	97.5
Dodecane	170.34	1770000	658.2	0.0164	121.6
D4	296.62	1330000	586.5	0.0157	127.6
PP5	462.00	1710000	565.2	0.0156	128.5
PP10	574.00	1580000	632.2	0.0128	156.2

For this report, the gas of interest is Perfluoroperhydrofluorene ($C_{13}F_{22}$), or PP10. Any particular gas in Tab. 1 could have been chosen; however, the large region in which $\mathcal{G} < 0$ for PP10 makes the gas favorable for good numerical results.

Many authors have investigated the effects of the $\mathcal{G} < 0$ region on aerodynamic quantities such as lift and drag, as well as in Rankine cycles [7 - 9]. The primary idea in this regime is to reduce elements of wave drag by reducing shock wave interactions while maintaining large temperature drops required for Rankine cycles. With much of the work in this area stemming from the past few decades, many of the useful benefits from the $\mathcal{G} < 0$ region are continuing to be explored by those hoping to take advantage of favorable wave dynamics on a system.




Figure 2: A non-convex, cubic function $f(x) = x^3$

curve [12] given by,

$$\kappa = \frac{f''(x)}{(1 + f'(x)^2)^{3/2}}, \quad (5)$$

where the term of importance is the second derivative of the function, $f''(x)$. At any point




Figure 3: Second derivatives of convex and non-convex functions.

should the second derivative of the function be negative, it is a non-convex function. To examine this property, the second derivatives of two of the before-mentioned examples are given in Fig. 3(a) and 3(b). Thus, any function with a negative second derivative will be considered non-convex. A given convexity plays a large role in wave formation and motion. Change of convexity is the cause of splitting of wave forms into fans and discontinuities, as will be seen through the examination of shock tube problems.

4 Governing Equations

The Euler equations are given with mass conservation as Eq. (6), linear momentum conservation as Eq. (7), and energy conservation as Eq. (8). The coupled van der Waals equation of state [13] is given in Eq. (9), and the caloric equation of state is given by Eq. (10) with a relation for density to specific volume in Eq. (11):

$$\frac{\partial \rho}{\partial t} + \nabla \cdot (\rho \mathbf{u}) = 0, \quad (6)$$

$$\frac{\partial(\rho \mathbf{u})}{\partial t} + \nabla \cdot (\rho \mathbf{u} \mathbf{u} + P) = \mathbf{0}, \quad (7)$$

$$\frac{\partial(\rho e + 1/2 \rho \mathbf{u} \cdot \mathbf{u})}{\partial t} + \nabla \cdot \rho \mathbf{u} \left(e + \frac{1}{2} \mathbf{u} \cdot \mathbf{u} + \frac{P}{\rho} \right) = 0, \quad (8)$$

$$P(T, v) = \frac{RT}{v - b} - \frac{a}{v^2}, \quad (9)$$

$$e = c_v T - a\rho, \quad (10)$$

$$v = \frac{1}{\rho}, \quad (11)$$

where ρ is the density, \mathbf{u} is the vector velocity of the fluid, and e is the specific internal energy. Note, for a van der Waals gas with constant c_v , the specific heat at constant pressure, c_P , is not a constant; hence, the gas is not calorically perfect. Further, in this analysis, we will restrict ourselves to the one-dimensional case. As a result, we possess six equations in one-

dimension with six unknowns (namely density, momentum, energy, temperature, velocity, and pressure), granting the ability to solve the system. In Eq. (9), the values of a and b are constants that modify the typical ideal gas equation to account for finite molecular volume of fluid particles and inter-molecular interactions between particles, and are chosen such that the experimentally observed critical point is a point of inflection for the critical isotherm in the $P - v$ plane. In canonical form for a standard van der Waals gas, the constants a and b are given by

$$a = \frac{27 R^2 T_c^2}{64 P_c}, \quad (12)$$

and

$$b = \frac{1 R T_c}{8 P_c}, \quad (13)$$

where in Eq. (12) and Eq. (13), T_c and P_c are the critical temperature and critical pressure, respectively. Values of a and b can be computed for each fluid in Tab. 1, and for PP10, values of a and b are $22.3889 \text{ m}^3/\text{kg}$ and $0.0007244 \text{ Pa m}^6/\text{kg}^2$, respectively.

To compute the adiabatic sound speed, the following approach was taken [13], starting from the differential Gibbs equation, one has

$$T ds = de + P dv. \quad (14)$$

To rewrite dv in terms of $d\rho$, Eq. (11) is differentiated to yield $dv = -1/\rho^2 d\rho$. Substituting this relation into Eq. (14), one finds

$$T ds = de - \frac{P}{\rho^2} d\rho. \quad (15)$$

In our case, BZT vapors may be modeled as simple compressible substances with $e = e(\rho, P)$, de may be expressed as

$$de = \left(\frac{\partial e}{\partial \rho} \right)_P d\rho + \left(\frac{\partial e}{\partial P} \right)_\rho dP. \quad (16)$$

Substituting Eq. (16) into Eq. (15), taking $ds = 0$, and solving for $\partial P/\partial\rho|_s$ produces

$$\left(\frac{\partial P}{\partial\rho}\right)_s = \frac{-\left(\frac{\partial e}{\partial\rho}\right)_P + \frac{P}{\rho^2}}{\left(\frac{\partial e}{\partial P}\right)_\rho}. \quad (17)$$

It can be shown that the adiabatic sound speed is the square root of the left-hand side of Eq. (17). Using Eq. (10) and substituting temperature for pressure by Eq. (9), the adiabatic sound speed for the van der Waals gas is found as

$$c = \sqrt{\frac{2abc_v - a(c_v - R)v + P(c_v + R)v^3}{c_v(v - b)v}}. \quad (18)$$

Note, if a and b are taken to 0, the ideal gas solution is recovered. With the adiabatic sound speed as computed in Eq. (18), it is possible to now calculate the fundamental derivative of gas dynamics by Eq. (2). Starting from Eq. (14) and integrating for a van der Waals fluid, one finds,

$$s - s_o = c_v \ln\left(\frac{T}{T_o}\right) + R \ln\left(\frac{v - b}{v_o - b}\right), \quad (19)$$

where T_o , v_o , and s_o are arbitrary reference states. Eq. (9) can be solved for temperature (also known as the thermal equation of state) and substituted into Eq. (19) for $s(P, v)$. Rearranging, one produces $P(s, v)$. The equation is given by

$$P(s, v) = \left(\frac{RT_o}{v_o - b}\right) \exp\left(\frac{s - s_o}{c_v}\right) \left(\frac{v_o - b}{v - b}\right)^{1+R/c_v} - \frac{a}{v^2}. \quad (20)$$

Note, if the gas is calorically perfect and ideal under an isentropic process, Eq. (20) reduces to $Pv^\gamma = \text{constant}$, where γ is the ratio of specific heats c_P/c_v . Eq. (2) requires $\partial^2 P/\partial v^2|_s$, and Eq. (20) is in the form necessary to compute the derivatives. Carrying out this process, substituting back in for P and ρ by Eq. (19) and Eq. (9) with $v = 1/\rho$, and substituting

Eq. (18) for the scaling factor, the fundamental derivative for van der Waals is found as

$$\mathcal{G}(P, \rho) = \frac{c_v \left(b - \frac{1}{\rho} \right) \left(-6a\rho^4 + \frac{(c_v+R)(2c_v+R)\rho^2(P+a\rho^2)}{c_v^2(-1+b\rho)^2} \right)}{(-2P\rho(c_v + R) + 2a\rho^3(c_v - R - 2bc_v\rho))}. \quad (21)$$

In the calorically perfect ideal gas limit, the fundamental derivative reduces to

$$\mathcal{G} = \frac{\gamma + 1}{2}, \quad (22)$$

For calorically perfect ideal gases, \mathcal{G} must be greater than unity. In regions where $0 < \mathcal{G} < 1$, non-ideal effects due to the van der Waals equation of state are found [5], but are not important for this study, as compression shocks and rarefaction fans are predicted in this range of \mathcal{G} .

As discussed, the $\mathcal{G} = 0$ curve is of particular importance for wave dynamics. Substituting our parameters for PP10 from Tab. 1, with R given by \hat{R}/M where \hat{R} is the universal gas constant, and calculating the necessary unit conversions, a curve for $\mathcal{G} = 0$ is produced and is shown in Fig. 4. As seen in Fig. 4, isentropes nearly coincide with isotherms for large c_v gases. As also demonstrated in Fig. 4, there is a change of convexity of the isentropes and isotherms outside of the $\mathcal{G} = 0$ curve to within the $\mathcal{G} = 0$ curve. Isentropes and isotherms under the $\mathcal{G} = 0$ curve experience a non-convex shape that is particularly important in the formation of a wave form.

5 Shock Capturing Methods

For the remainder of this paper, solutions will be provided to shock tube problems governed by the Euler equations and different sets of initial conditions. The Euler equations are conservation laws of the form:

$$\frac{\partial u}{\partial t} + \frac{\partial f(u)}{\partial x} = 0, \quad (23)$$




Figure 4: A plot of $\mathcal{G} = 0$, PP10's critical isentrope, and a family of isotherms.

where u is the conserved variable, and $f(u)$ is the flux of the conserved variable. The convexity of the flux function indicates changes in the shape of a wave form, and more in-depth analysis of this phenomenon in the scalar sense is given in Davies [10]. One first-order approach to solving conservation laws of this form is the Lax-Friedrichs finite differencing method [14], provided as:

$$u_j^{n+1} = \frac{1}{2}(u_{j-1}^n + u_{j+1}^n) - \frac{\Delta t}{2\Delta x}(f(u)_{j+1}^n - f(u)_{j-1}^n). \quad (24)$$

The Lax-Friedrichs method is a forward differencing scheme in time and a central-differencing scheme in space that is rudimentary, but useful for shock tube problems. For a sample Lax-Friedrichs code in the scalar case, one can refer to Davies [10]. Often common among first-order methods, the Lax-Friedrichs method is rather dissipative near discontinuities as

is demonstrated in Fig. 5.

For the purpose of this paper, the numerical viscosity inherent to Lax-Friedrichs hides important results. For initial conditions in which we should expect anomalous behavior of waves (i.e. discontinuous rarefactions/compressions, isentropic compressions/rarefactions, or some combination), it is important that a distinction can be made between physically continuous wave forms and numerically induced dissipation. To address this issue, the second-order Richtmyer two-step Lax-Wendroff method is useful. The two-step Lax-Wendroff method is given as [14],

$$\begin{aligned} u_{j+1/2}^{n+1/2} &= \frac{1}{2}(u_j^n + u_{j+1}^n) - \frac{\Delta t}{2\Delta x}(f(u_{j+1}^n) - f(u_j^n)), \\ u_j^{n+1} &= u_j^n - \frac{\Delta t}{\Delta x}(f(u_{j+1/2}^{n+1/2}) - f(u_{j-1/2}^{n+1/2})). \end{aligned} \tag{25}$$

The two-step Lax-Wendroff method estimates the value of u at the half step $u_{j+1/2}^n$, and uses this to perform central differencing about the $j+1/2$ spatial grid point to gather a value for u at the $n+1/2$ time step. A similar process is used for the $u_{j-1/2}^n$ point, as a central-difference is simply taken about $j-1/2$. A full time step is then taken by taking a step of Δt with a central difference at the j grid point. Common among nominally second order methods, the two-step Lax-Wendroff method demonstrates oscillations near discontinuities. Though some numerical dissipation is present, it is not nearly as prevalent as it is in nominally first-order methods, and for this reason, solutions in this paper are limited to the Lax-Wendroff method.

An air ($R = 287$ J/kg/K, $c_v = 717.5$ J/kg/K), ideal gas, Sod shock tube problem with left side initial conditions, $P_L = 500000$ Pa and $v_L = 0.3876$ m³/kg and right side initial conditions $P_R = 101325$ Pa and $v_R = 0.8163$ m³/kg is provided to demonstrate a comparison between the Lax-Friedrichs method, Lax-Wendroff method, and an exact solution. This comparison is shown in Fig. 5.

As is common among all shock capturing methods, the two-step Lax-Wendroff method converges at less than first order due to large differences in solutions at discontinuity locations. An L_1 error analysis comparing numerical solutions to the exact solution in Fig. 5 was




Figure 5: A plot comparing the Lax-Friedrichs method, Lax-Wendroff method, and an exact solution of an ideal gas Sod shock tube problem with $\Delta x = 1 \times 10^{-3}$ m, $\Delta t = 1 \times 10^{-6}$ s at $t = 0.0005$ s.

completed on the two-step Lax-Wendroff method. The result of this error analysis is given in Fig. 6. In Fig. 6, the exponent of Δx indicates the order of convergence of the method. The L_1 error was taken by summing the scaled density error over the entire domain, and dividing by the grid size. This is represented by Eq. (26):

$$L_1 \rho \text{ Error} = \left(\sum_{i=1}^{N_{\text{grid}}-1} \frac{|\rho_{\text{exact},i} - \rho_{\text{LW},i}|}{\rho_{\text{exact},i}} \right) \frac{1}{N_{\text{grid}} - 1}. \quad (26)$$

In Eq. (26), ρ_{exact} indicates density of the exact solution, ρ_{LW} indicates the density of the numerical solution, and N_{grid} represents the number of grid points in the domain. For each grid, the number of points in the exact solution was adjusted such that it corresponded in size to the numerical solution.

For the remainder of this report, solutions are run at a grid size of $\Delta x = 3 \times 10^{-5}$ m for a total of 33333 points over the domain. The value of Δt was determined by the maximum




Figure 6: Scaled L_1 density error as a function of grid size.

expected wave speed for the given problem. As BZT gases are typically slower propagating gases, the expected wave speed is low. For stability, Δt was encoded as $\Delta x/(1000 \text{ m/s})$, as the maximum wave speed to “outrun” the grid is projected to be much less than 1000 m/s. A different, perhaps better, approach for future implementation involves scanning the grid for the largest value of the wave speed, $u + c$, and setting the value of Δt accordingly. The Lax-Wendroff code used to create the results here is given in Appendix A.

6 The Convex Case

The first case of particular interest is the case in which $\mathcal{G} > 0$. For all cases, the wave motion is driven by a reservoir of high pressure and density that drives the flow when a “diaphragm” is released that allows for the interaction between both sets of initial conditions. The values for the initial conditions in this case and the cases to follow are provided in Tab 2, and important parameters regarding the PP10 gas necessary for shock tube initial conditions is provided in Tab. 3. The left-side initial pressure is given as P_L , and the specific volume is

given by v_L . The right side initial conditions are also provided and are given with subscript R . In this particular example, both initial conditions are located in the regime where $\mathcal{G} > 0$. A graphical depiction is given in Fig. 7. As predicted by the literature, typical behavior of gas dynamics should be expected in this region of $P - v$ space. Solutions for the set of initial conditions described above are provided in Fig. 8.




Figure 7: A plot of $\mathcal{G} = 0$, PP10's critical isentrope, a family of isotherms, and two points representing initial conditions separated by a diaphragm for the convex case.

Table 2. A summary of initial conditions used for solutions to Sod shock tube problems.

Case	P_L (Pa)	v_L (m^3/kg)	P_R (Pa)	v_R (m^3/kg)
Convex I.C.'s	2.5×10^6	0.0022	2.0×10^6	0.0035
Non-Convex I.C.'s	1.74×10^6	0.0027	1.5×10^6	0.0035
Mixed-Convexity I.C.'s	2.0×10^6	0.0022	1.5×10^6	0.0035

In Fig. 8, it becomes apparent the rightwards progressing wave form propagates as a




Figure 8: The shock tube problem with initial conditions located in the convex isentropic region for $t = 0.005$ s.

discontinuous compression, or shock wave, and the leftwards propagating wave holds the shape of a continuous or isentropic compression. This is as expected, as the theory for $\mathcal{G} < 0$ agrees with observed behavior in general gas dynamics.

Table 3. A summary of gas values, PP10, for initial conditions in the shock tube.

R (J/kg·K)	c_v (J/kg·K)	a (m ⁵ /kg·s)	b (m ³ /kg)
14.4843	1131.588	22.3889	0.0007244

7 The Non-Convex Case

The non-convex case introduces initial conditions such that both sets are located under




Figure 9: A plot of $\mathcal{G} = 0$, PP10's critical isentrope, a family of isotherms, and two points representing initial conditions separated by a diaphragm for the non-convex case.

the $\mathcal{G} = 0$ curve. In this region of $P - v$ space, anomalous behavior is predicted. The set of initial conditions as provided is presented graphically in Fig. 9.




Figure 10: The shock tube problem with initial conditions located in the non-convex isentrope region for $t = 0.005$ s.

As the initial conditions are located above the critical isotherm, it can be stated with certainty that the fluid is a single-phase vapor. Solutions to the non-convex case are presented in Fig. 10. As is seen in Fig. 10, the rightwards progressing wave traverses as a continuous or isentropic compression, and the leftwards traveling wave progresses as a discontinuous rarefaction. Note that the contact discontinuity remains intact, and only impacts the density and temperature solutions as expected. Based off of the presented behavior, we can say with certainty that the speed of sound's relation with state variables has reversed from normally expected gas behavior. To ensure the second law's satisfaction, a plot of entropy is provided in Fig. 10(e). In order to satisfy the second law of thermodynamics, it must be seen that entropy remains constant, or increases in order to demonstrate plausibility in nature. As demonstrated in Fig. 10, this is the case. Note, Fig. 10(e) was generated as an exact solution to the Rankine-Hugoniot jump conditions [2], as the discretization required to visually demonstrate an entropy increase on the order of 10^{-4} with the two-step Lax-Wendroff method is impractical. Largely this is the case due to the weak nature of the rarefaction shocks, as the differences in pressure in the $\mathcal{G} < 0$ range are relatively small in comparison to the convex region. Due to the large mass of the fluid and smaller differences in pressure, velocities and sound speeds are typically low, and entropy changes are present, but small.

8 The Case of Mixed Convexity

The last set of initial conditions of interest is the case of mixed convexity. This set of initial conditions is given by the left-side initial conditions positioned in the convex region, or $\mathcal{G} > 0$ region, and the right-side initial conditions positioned in the $\mathcal{G} < 0$ regime. A graphical depiction is presented in Fig. 11. In this regime, we should expect the pressure to drive the flow into the non-convex region where $\mathcal{G} < 0$. In this particular case, we should expect some combination of ideal compressible flow and anomalous behavior, as one wave






Figure 11: A plot of $\mathcal{G} = 0$, PP10's critical isentrope, a family of isotherms, and two points representing initial conditions separated by a diaphragm for the mixed convexity case.


will be traversing away from the anomalous region, and the other wave will traverse into it. The behavior demonstrated from this particular set of initial conditions is provided in Fig. 12. Note, the contact discontinuity is visible in both density and temperature solutions, yet it is absent in the pressure solution as expected. The fundamental derivative changes sign at around $x = 0.2$ m, and this is apparent in the pressure, temperature, and density solutions. As the sign of the fundamental derivative changes, the waveform progresses from a discontinuous compression, or shock wave, into an isentropic or continuous compression marked by the anomalous zone. The left-traveling wave traverses as a continuous rarefaction typical of general gas behavior. It thus becomes obvious that the sign of the fundamental derivative plays an indisputable role in the form of a real gas waveform. Assuming that left-side initial conditions are greater in magnitude than right-side conditions, a summary of gas behavior as it relates to the sign of the fundamental derivative is provided in Tab. 4 for




(a) Density




(b) \mathcal{G}




(c) Pressure



(d) Temperature



(e) Entropy near the split shock



(f) Speed of Sound

Figure 12: The shock tube problem with mixed convex and non-convex isentrope located initial conditions for $t = 0.005$ s.

reference.

Table 4. A summary of wave formation as it relates to the sign of the fundamental derivative, \mathcal{G} , assuming the left initial condition is at higher P , ρ , and T .

Sign of \mathcal{G}	Left-Traveling Wave	Right-Traveling Wave
$\mathcal{G} > 0$	Continuous Rarefaction	Shock Wave
$\mathcal{G} < 0$	Discontinuous Rarefaction	Isentropic Compression
Straddling I.C.'s	Continuous Rarefaction	Split Shock, Isentropic Compression Wave

The speed of sound's relations to the state variables reverses at the same $x = 0.2$ m location. This explains the split in the wave form, as characteristics are converging into an infinitesimally small shock wave in the convex region, but diverge with the speed of sound's decrease in the non-convex region. For the entropy as given in in Fig. 12(e), the Rankine-Hugoniot jump equations predict an increase in entropy of 0.0023 J/kg/K, and the numerical method demonstrates an entropy increase of 0.0024 J/kg/K.

9 Discussions and Conclusions

Solutions to shock tube problems for the van der Waals gas in the BZT region are presented and analyzed. As agrees with the literature, anomalous behavior is seen in regions where the fundamental derivative is negative, as the speed of sound's relation to the state variables of P , T , and ρ is reversed and behaves opposite of what is generally predicted in typical gas dynamics. Some famous works [15 - 17], perhaps in a restriction to the ideal gas, note an impossibility of rarefaction shock waves due to their instability as it relates to the speed of sound. Though described in many different ways, it is mainly explained that state information behind the rarefaction wave must propagate faster than the wave speed of the rarefaction shock and result in the destruction of the wave form. However, it has been shown that rarefaction shock waves are stable and possible in regions where the speed of sound's relation to state variables is reversed, as is the case in the BZT region. As was shown, for

initial conditions located in the convex region, typical gas dynamics remained. For the case of non-convex initial conditions in which the magnitude of the left-side dominated, an isentropic compression propagated rightwards, and a discontinuous rarefaction was seen moving leftwards, and in the case of mixed convexity, a split shock and continuous compression was demonstrated with expected behavior elsewhere.

Future work in this category may be extended to BZT flows over an aerodynamic body such as the classic wedge problem. For this experiment, the methods as before described could be extended to two-dimensions for application. As a result, analysis may be extended to oblique shock waves and expansions and their effects which may continue to reverse typical thinking and encourage practical application of rarefaction shock waves and isentropic compressions in practice.

References

- [1] Sod, G.A., 1978, A Survey of Several Finite Difference Methods for Systems of Nonlinear Hyperbolic Conservation Laws. *Journal of Computational Physics*, Vol. 27, pp. 1-31.
- [2] Bethe, H.A., 1942, The Theory of Shock Waves for an Arbitrary Equation of State, Technical Report 545, Office of Scientific Research and Development, Los Alamos National Laboratory, New Mexico.
- [3] Zel'dovich, Y.B., 1946, On the Possibility of Rarefaction Shock Waves, *Zhurnal Eksperimental'noi i Teoreticheskoi Fiziki*, Vol. 16, pp. 363-364, (Translated in: 1960, Joint Publications Research Service 7320, Arlington, Virginia.).
- [4] Lambrakis, K., and Thompson, P.A., 1972, Existence of Real Fluids with a Negative Fundamental Derivative Γ , *Physics of Fluids*, Vol. 5, pp. 933-935.
- [5] Colonna, P., and Guardone, A., 2006, Molecular Interpretation of Nonclassical Gas Dynamics of Dense Vapors under the van der Waals Model, *Physics of Fluids*, Vol. 18, 056101.
- [6] Thompson, P.A., A Fundamental Derivative in Gasdynamics, *Physics of Fluids*, Vol. 14, pp. 1843-1849.
- [7] Cramer, M.S., and Tarkenton, G.M., 1992, Transonic Flows of Bethe-Zel'dovich-Thompson Fluids, *Journal of Fluid Mechanics*, Vol. 240, pp. 197-228.
- [8] Cinnella, P., and Congedo P.M., 2005, Aerodynamic Performance of Transonic Bethe-Zel'dovich-Thompson Flows Past an Airfoil, *AIAA Journal*, Vol. 43, pp. 370-378.
- [9] Congedo, P.M., Corre, C., and Cinnella P., 2007, Airfoil Shape Optimization for Transonic Flows of Bethe-Zel'dovich-Thompson Fluids, *AIAA Journal*, Vol. 45, pp. 1303-1316.

- [10] Davies, A.M., 2020, *Numerical Solutions to Scalar Conservation Laws*, Department of Aerospace and Mechanical Engineering, University of Notre Dame, Notre Dame, Indiana, pp. 6-8. <https://www3.nd.edu/~powers/>
- [11] Borwein, J., and Lewis, A., 2000, *Convex Analysis and Nonlinear Optimization*, (Springer, New York). pp. 1.
- [12] Kline, M., 1998, *Calculus: An Intuitive and Physical Approach*, (Courier Corporation, North Chelmsford). pp. 457–461.
- [13] Powers, J. M., 2020, *Lecture Notes on Thermodynamics*, Department of Aerospace and Mechanical Engineering, University of Notre Dame, Notre Dame, Indiana. <https://www3.nd.edu/~powers/ame.20231/notes/notes.pdf>
- [14] LeVeque, R.J., 1992, *Numerical Methods for Conservation Laws*, (Birkhäuser Basel).
- [15] Shapiro, A.H., 1953, *The Dynamics and Thermodynamics of Compressible Fluid Flow*, (John Wiley and Sons, New York).
- [16] Liepmann, H.W., and Roshko, A., 2013, *Elements of Gas Dynamics*, (Dover Publications, New York).
- [17] Zel'dovich, Y.B., 2012, *Physics of Shock Waves and High-Temperature Hydrodynamic Phenomena*, (Dover Publications, New York).

Appendix A

A FORTRAN 90 code used for the Richtmyer two-step Lax-Wendroff method to solve the inviscid, compressible Euler equations is provided below:

```
program EulerEquation
  use caloricequationofstateLW
  use thermalequationofstateLW
  use pressuresLW
implicit none
  real (kind=8):: dx,dt,ttot,xtot,pi,&
                    gam,rhoR,PR,PL,rhoL,v0,cv,Tc,Pc,&
                    R,a,b,o,vL,vR,r12,rm12,m12,&
                    mm12,e12,em12,p12,pm12
  integer:: N,K,i,j,eos,savestep,Nhold

  real*8,allocatable:: x(:),ti(:),rho(:,:),m(:,:),P(:,:),&
  e(:,:),T(:,:)
  open(12,file='initialconditions.in')
  read(12,100) dx
  dt=dx/1000
  read(12,100) ttot
  read(12,100) xtot
  read(12,100) pi
  N=ttot/dt
  K=xtot/dx
  read(12,101) savestep
  read(12,100) o
```

```

read(12,100) gam
read(12,100) vL
rhoL=1/vL
read(12,100) PR
read(12,100) PL
read(12,100) vR
rhoR=1/vR
read(12,100) v0
read(12,100) cv
read(12,100) R
read(12,100) a
read(12,100) b
! read(12,100) Pc
! read(12,100) Tc
100 format(19x,f12.7)
101 format(19x,i10)

if (savestep == 0) then
allocate(x(K))
allocate(ti(N))
allocate(rho(K,2))
allocate(m(K,2))
allocate(P(K,2))
allocate(T(K,2))
allocate(e(K,2))
! allocate(u(K,2))

```

```

else
allocate(x(K))
allocate(ti(N))
allocate(rho(K,N))
allocate(m(K,N))
allocate(P(K,N))
allocate(T(K,N))
allocate(e(K,N))
! allocate(u(K,N))
end if

do i=1,K
    x(i)=-0.5+dx*(i-1)
end do

do i=1,N
    ti(i)=-0+dt*(i-1)
end do

Nhold = N ! To set limits for do loops
if (savestep == 0) then
    N=2
else
    continue
end if

```



```

do j=1,K
  if (x(j).lt.0) then
    rho(j,1)=rhoL
    m(j,1)=rhoL*v0
    P(j,1)=PL
    call teos(rho(j,1),P(j,1),R,a,b,T(j,1))
    call ceos(dt,dx,m(j,1),DBLE(0.0),P(j,1),&
      DBLE(0.0),rho(j,1),&
      DBLE(0.0),DBLE(0.0),DBLE(0.0),&
      DBLE(0.0),cv,R,a,b,0,e(j,1))

  else
    rho(j,1)=rhoR
    m(j,1)=rhoR*v0
    P(j,1)=PR
    call teos(rho(j,1),P(j,1),R,a,b,T(j,1))
    call ceos(dt,dx,m(j,1),DBLE(0.0),&
      P(j,1),DBLE(0.0),rho(j,1),&
      DBLE(0.0),DBLE(0.0),DBLE(0.0),&
      DBLE(0.0),cv,R,a,b,0,e(j,1))

  end if
end do

do i=1,N
  rho(K,i)=rhoR
  m(K,i)=rhoR*v0

```

```

P(K,i)=PR
call teos(rho(K,i),P(K,i),R,a,b,T(K,i))
call ceos(dt,dx,m(K,i),&
          DBLE(0.0),P(K,i),DBLE(0.0),rho(K,i),&
          DBLE(0.0),DBLE(0.0),&
          DBLE(0.0),DBLE(0.0),cv,R,a,b,0,e(K,i))

! u(K,i)=m(K,i)/rho(K,i)
end do

do i=1,N
  rho(1,i)=rhoL
  m(1,i)=rhoL*v0
  P(1,i)=PL
  call teos(rho(1,i),P(1,i),R,a,b,T(1,i))
  call ceos(dt,dx,m(1,i),&
            DBLE(0.0),P(1,i),DBLE(0.0),rho(1,i),&
            DBLE(0.0),DBLE(0.0),&
            DBLE(0.0),DBLE(0.0),cv,R,a,b,0,e(1,i))

! u(1,i)=m(1,i)/rho(1,i)
end do

do i=1,Nhold-1
  do j=2,(K-1)

```

```

if (savestep==0) then
  r12=(1.0d0/2.0d0)*(rho(j,1)+rho(j+1,1))-&
    (dt/2.0d0)/(dx)*(m(j+1,1)-m(j,1))
  rm12=(1.0d0/2.0d0)*(rho(j,1)+rho(j-1,1))-&
    (dt/2.0d0)/(dx)*(m(j,1)-m(j-1,1))

  m12=(1.0d0/2.0d0)*(m(j,1)+m(j+1,1))-dt/(2.0d0*dx)*&
    ((m(j+1,1)*m(j+1,1)/rho(j+1,1)+P(j+1,1))-&
    (m(j,1)*m(j,1)/rho(j,1)+P(j,1)))
  mm12=(1.0d0/2.0d0)*(m(j,1)+m(j-1,1))-&
    dt/(2.0d0*dx)*((m(j,1)*m(j,1)/rho(j,1)+P(j,1))-&
    (m(j-1,1)*m(j-1,1)/rho(j-1,1)+P(j-1,1)))

  e12=(1.0d0/2.0d0)*(e(j,1)+e(j+1,1))-&
    dt/(2.0d0*dx)*(((m(j+1,1)/rho(j+1,1))*&
    (e(j+1,1)+P(j+1,1)))-&
    ((m(j,1)/rho(j,1))*(e(j,1)+P(j,1))))
  em12=(1.0d0/2.0d0)*(e(j,1)+e(j-1,1))-&
    dt/(2.0d0*dx)*(((m(j,1)/rho(j,1))*&
    (e(j,1)+P(j,1)))-&
    ((m(j-1,1)/rho(j-1,1))*(e(j-1,1)+P(j-1,1))))

  call pressure(rm12,mm12,em12,cv,R,a,b,pm12)
  call pressure(r12,m12,e12,cv,R,a,b,p12)

  rho(j,2)=rho(j,1)-dt/dx*(m12-mm12)
  m(j,2)=m(j,1)-dt/dx*((m12*m12/r12+p12)-&

```

```

(mm12*mm12/rm12+pm12))
call ceos(dt,dx,m12,mm12,p12,pm12,r12,rm12,&
        e12,em12,e(j,1),cv,R,a,b,1,e(j,2))
call pressure(rho(j,2),m(j,2),e(j,2),cv,R,a,b,P(j,2))
call teos(rho(j,2),P(j,2),R,a,b,T(j,2))

else

r12=(1.0d0/2.0d0)*(rho(j,i)+rho(j+1,i))
r12=r12-(dt/2.0d0)/(dx)*(m(j+1,i)-m(j,i))
rm12=(1.0d0/2.0d0)*(rho(j,i)+rho(j-1,i))
rm12=rm12-(dt/2.0d0)/(dx)*(m(j,i)-m(j-1,i))

m12=(1.0d0/2.0d0)*(m(j,i)+m(j+1,i))-&
    dt/(2.0d0*dx)*(m(j+1,i)*&
    m(j+1,i)/rho(j+1,i)+P(j+1,i)-&
    (m(j,i)*m(j,i)/rho(j,i)+P(j,i)))
mm12=(1.0d0/2.0d0)*(m(j,i)+m(j-1,i))-&
    dt/(2.0d0*dx)*(m(j,i)*m(j,i)/rho(j,i)+P(j,i)-&
    (m(j-1,i)*m(j-1,i)/rho(j-1,i)+P(j-1,i)))

e12=(1.0d0/2.0d0)*(e(j,i)+e(j+1,i))-&
    dt/(2.0d0*dx)*((m(j+1,i)/rho(j+1,i))*&
    (e(j+1,i)+P(j+1,i))-&
    (m(j,i)/rho(j,i))*(e(j,i)+P(j,i)))
em12=(1.0d0/2.0d0)*(e(j,i)+e(j-1,i))-&
    dt/(2.0d0*dx)*((m(j,i)/rho(j,i))*(e(j,i)+P(j,i))-&
    (m(j-1,i)/rho(j-1,i))*(e(j-1,i)+P(j-1,i)))

```

```

call pressure(rm12,mm12,em12,cv,R,a,b,pm12)
call pressure(r12,m12,e12,cv,R,a,b,p12)

rho(j,i+1)=rho(j,i)-dt/dx*(m12-mm12)
m(j,i+1)=m(j,i)-dt/dx*&
(m12*m12/r12+p12-mm12*mm12/rm12+pm12)
call ceos(dt,dx,m12,mm12,p12,pm12,r12,rm12,&
e12,em12,e(j,i),cv,R,a,b,1,e(j,i+1))

    end if
end do

if (savestep==0) then
    rho(:,1)=rho(:,2)
    m(:,1)=m(:,2)
    e(:,1)=e(:,2)
    P(:,1)=P(:,2)
    T(:,1)=T(:,2)

else
    continue
end if

end do

open(7,file="rhoOutput.txt")
do i=1,K

```

```

        write(7,*) x(i), rho(i,N)
    end do
close(7)

! open(7,file="VelocityOutput.txt")
! do i=1,K
!     write(7,*) x(i), u(i,N)
! end do
! close(7)

open(7,file="PressureOutput.txt")
do i=1,K
    write(7,*) x(i), P(i,N)
end do
close(7)

open(7,file="TemperatureOutput.txt")
do i=1,K
    write(7,*) x(i), T(i,N)
end do
close(7)

end program EulerEquation

```

The modules presented in the code are given by:

```

module thermalequationofstateLW
contains
subroutine teos(rho,P,R,a,b,T)
implicit none
real (kind=8), intent(in) :: rho
real (kind=8), intent(in) :: P
real (kind=8), intent(in) :: R
real (kind=8), intent(in) :: a
real (kind=8), intent(in) :: b
real (kind=8), intent(out) :: T


$$T=(-1)*((-1+b*rho)*(P+a*rho**2))/(R*rho)$$


end subroutine teos
end module thermalequationofstateLW

```

```

module caloricequationofstateLW
contains
subroutine ceos(dt,dx,m12,mm12,p12,pm12,&
r12,rm12,e12,em12,em,cv,R,a,b,v,e)
implicit none
real (kind=8), intent(in) :: dt
real (kind=8), intent(in) :: m12
real (kind=8), intent(in) :: mm12

```

```

real (kind=8), intent(in) :: p12
real (kind=8), intent(in) :: pm12
real (kind=8), intent(in) :: r12
real (kind=8), intent(in) :: rm12
real (kind=8), intent(in) :: dx
real (kind=8), intent(in) :: e12
real (kind=8), intent(in) :: em12
real (kind=8), intent(in) :: em
real (kind=8), intent(in) :: cv
real (kind=8), intent(in) :: R
real (kind=8), intent(in) :: a
real (kind=8), intent(in) :: b
integer, intent(in) :: v
real (kind=8), intent(out) :: e

if (v==0) then

    e=r12*(cv/R)*(1/r12-b)*&
    (p12+a*r12**2)-a*r12**2+(0.5)*r12*(m12/r12)**2

else

e=em-dt/dx*(((m12/r12)*(e12+p12))-((mm12/rm12)*(em12+pm12)))

end if

end subroutine ceos

```



```

end module caloricequationofstateLW

module pressuresLW
contains
subroutine pressure(rho,m,e,cv,R,a,b,Pres)
implicit none
real (kind=8), intent(in) :: rho
real (kind=8), intent(in) :: m
real (kind=8), intent(in) :: e
real (kind=8), intent(in) :: cv
real (kind=8), intent(in) :: R
real (kind=8), intent(in) :: a
real (kind=8), intent(in) :: b
real (kind=8), intent(out) :: Pres

Pres=(R/cv)*(1/(1-rho*b))*&
((e-(1./2.)*m*m/rho)+a*rho**2)-a*rho**2

end subroutine pressure
end module pressuresLW

```

The file of initial conditions is provided as:

```
dx = 3.0d-6
```

ttot = 5.0d-3
xtot = 1.0d0
pi = 3.14159265d0
savestep = 0
o = 0
gam = 1.4d0
vL = 0.0022d0
PR = 2000000.0d0
PL = 2500000.0d0
vR = 0.0035d0
v0 = 0.0d0
cv = 1131.5875d0
R = 14.484d0
a = 22.3889d0
b = 0.0007244d0